T65-4-1

COMPUTER-AIDED ANTENNA TUNER DESIGN

Mark Connelly -- WA1ION DX Labs -- 31 DEC 1984

Numerous passive and active receiving-antenna tuners have have been developed recently: most embody the idea of simplicity and ease of construction. Yaesu, MFJ, McKay-Dymek, and Grove have all offered such devices commercially to shortwave and medium-wave DX enthusiasts. A common practice is to use one tuning capacitor (often the common 10 to 365 pF air-variable) and numerous inexpensive molded inductors generally mounted on a ganged rotary switch which selects the various tuning ranges. The use of inductive-dividers allows the impedance-transfer from high (at the tank circuit) to low/medium (at the tuner's output).

Although higher Q and greater output may sometimes be obtained with a tuner using hand-wound coils (having link windings and tapped main windings) and using two or more variable capacitors (e. g. Bob Foxworth's excellent passive tuner of early '70s vintage), experience has shown that even the best passive tuner cannot match the Q and gain of an active, regenerative tuning unit.

Aside from specialised cases demanding super Q and gain, there is a considerable need for moderate Q tuners having the benefits of low cost, simplicity of construction, and ease of operation.

The urban DXer, for instance, has a great difficulty with spurious responses (resulting from the poor strong-signal handling characteristics of many of today's receivers) when any wire more than a few feet in length is connected to the antenna jack. "Spurs" noted may be caused by AM, FM, & TV stations as well as by local utility, amateur, & CB transmitters. Even in rural locations, some receivers may generate images & spurs (chiefly because of overloading by potent international shortwave broadcast stations & RTTY/FAX/CW utility stations).

A tuner (pre-selector) goes a long way to eliminate these problems. "De facto" gain is accomplished even with a somewhat-lossy, loosely-coupled passive tuner at urban sites because considerably-longer aerials may be used before spurious signals become a problem. With an active tuner of relatively simple design, aerials as short as car whips can provide worthwhile DX: this is an important consideration for the space-limited DXer in the city.

Two basic designs were chosen to be modelled for computer-analysis - one passive and one active. The passive design is represented by the schematic of Figure 1.

Figure 1



The coils shown (Lcoupling, Lmain, & Ltap) are actually coil-groups mounted on a rotary switch. Figure 2 illustrates this idea.

Figure 2



Experimentation has shown that good passive-tuner results are obtained with a wide variety of input antenna lengths and output load impedances if the coupling inductor (for loosely coupling inductively-reactive antennae) is approximately 3.5 to 5 times the value of the main inductor and if the main inductor is about 5 to 8 times the value of the tap (output) inductor. Increasing these ratios results in greater signal loss (although, perhaps, higher Q) whereas decreasing the ratios reduces the varieties of input & output impedances with which the tuner will work.

The coupling capacitor chosen (for loosely coupling capacitively-reactive antennae) should be about 7% to 20% of the maximum value of the tuning capacitor. 47 pF is a good overall coupling-capacitor choice to be used with a 10 to 365 pF tuning cap.; you might want to go a bit lower for active tuner applications (e. g. 33 pF) and a bit higher (e. g. 62 pF) for longwave (F < 500 kHz) passive tuners.

Figure 3 illustrates the active tuner circuit being considered for computer-aided design. Note that the tap inductor group plays an input, rather than an output, role (no separate "Lcoupling" group is used).

Figure 3



T65-4-2

The TUNER program - how it works:

In modelling the circuits of Figures 1 & 3, several assumptions were made:

1. We know the overall frequency range desired.

2. We know the minimum & maximum capacitance value of the variable capacitor to be used.

3. Entry of stray capacitance should be possible. (This is a ballpark figure entered to account for capacitance from tank-high to chassis ground caused by component layout & wiring.)

4. We know the value of the input coupling capacitor (typically 12% of the tuning cap.'s maximum value).

5. Inductors chosen by the program should be standard-value components (e. g. 10 uH, 220 uH). Coupling inductors are to be approximately 4 times the values of the corresponding main inductors; main inductors are to be approximately 6 times the values of the corresponding tap inductors.

6. Frequency-ranges should overlap slightly from one coil-switch position to the next even with worst-case input & output loading conditions. Nevertheless, the program should endeavour to use the fewest number of inductors and coil-switch positions required to make a working tuner design.

7. Desired output should be written to a text file on disk; such a file could be edited with suitable word-processing software (e. g. EDT), merged with other files, printed to paper or CRT, etc. [For those without this file-management facility, minor modification of the program will enable direct output to a lineprinter].

8. Desired output should display all input parameters (1, 2, 3, & 4 above) and the output of inductor values chosen for active & passive models (accompanied by full frequency-range data for L-coupling & C-coupling).

9. Minimum & maximum tuning capacitance should be adjusted somewhat
   (a) to allow for slight movement of the capacitor beyond the rated end-of-range frequency position, and
   (b) to compensate for variations in actual tuning capacitor values from their published specifications.
   The program's formula calculates a "useful" range of ((365-10)*0.1)+10, or 45.5 pF minimum to (365*0.8)=292 pF maximum for the standard 10 to 365 pF variable capacitor. Such adjustment reduces the likelihood of gaps between tuning ranges.

10. Worst-case minimum frequencies are those made the highest by the characteristics of the input & output impedances; worst-case maximum frequencies are those made the lowest by the characteristics of the input & output impedances.

    Simple circuit math shows that, in the passive tuner case:
    Worst-case min. freq. (L-coupled) occurs with a low-Z (approx. shorted) input & output.
    Worst-case min. freq. (C-coupled) occurs with a high-Z (approx. open) input & a low-Z (approx. shorted) output.
    Worst-case max. freq. (L-coupled) occurs with a high-Z (approx. open) input & output.
    Worst-case max. freq. (C-coupled) occurs with a low-Z (approx. shorted) input & a high-Z (approx. open) output.

    For active tuners, output impedance is not an important frequency-range-determining consideration because of the buffering provided by the FET-input amplifier circuit. Input (antenna) loading sets the worst-case conditions for the active tuner:
    Worst-case min. freq. (L-coupled) occurs with a low-Z (approx. shorted) input.
    Worst-case min. freq. (C-coupled) occurs with a high-Z (approx. open) input.
    Worst-case max. freq. (L-coupled) occurs with a high-Z (approx. open) input.
    Worst-case max. freq. (C-coupled) occurs with a low-Z (approx. shorted) input.

11. The program should be written in a manner that, upon examining the remarks in the listing, circuit-math formulae & design considerations used should be evident to anyone who has even a relatively-limited electronics background.

Program Listing

The following program is written in DEC PDP-11 BASIC-PLUS-2 V2.1-00.

(begin listing)

```
 1 PRINT\PRINT "TUNER.B2S"\PRINT"Receiving Tuner Design Aid"
 2 PRINT "WALION DX Labs / Software Division / 31 DEC 1984"\PRINT
 3 DIM V(12),L(12),F(4),C(4)
 4 K=159154.94\REM (1000000/(2*PI))
 5 U2$=""
 6 FOR I=1 TO 72\U2$=U2$+"*"\NEXT I
12 REM ** STANDARD INDUCTOR VALUES **
13 V(1)=1.2
14 V(2)=1.5
15 V(3)=1.8
16 V(4)=2.2
17 V(5)=2.7
18 V(6)=3.3
19 V(7)=3.9
20 V(8)=4.7
21 V(9)=5.6
22 V(10)=6.8
23 V(11)=8.2
24 V(12)=10
26 PRINT\INPUT "DATA OUTPUT FILE NAME";S$
27 OPEN S$ FOR OUTPUT AS FILE #1%
28 D$="A"
29 J=13\M=6\T=0\Z=0
30 IF D$="P" THEN PRINT #1,U2$\GO TO 50
32 INPUT "Minimum frequency, kHz";FMIN
33 INPUT "Maximum frequency, kHz";FMAX
35 INPUT "(40 pF typical)  Stray C, pF";CS
37 INPUT "Input coupling capacitor, pF";CC
38 CSH=CC+CS\REM shunt-C for C-coupling
39 INPUT "Minimum value of tuning capacitor, pF";CMIN
40 INPUT "Maximum value of tuning capacitor, pF";CMAX
41 REM +++ Practical min. Ctune will be set to CMIN+(0.1*(CMAX-CMIN)) +++
42 CMIN2=CMIN+(0.1*(CMAX-CMIN))
43 REM +++        Use 80% of maximum tuning capacitance                   +++
44 REM +++        to account for component variation.                     +++
45 CMAX2=0.8*CMAX
46 C(4)=CMIN2+CS\REM   min. equiv. C                        (L-coupling)
47 C(2)=CMIN2+CSH\REM  min. equiv. C with shorted input     (C-coupling)
48 C(1)=CMAX2+CS\REM   max. equiv. C with open input        (C-coupling)
49 C(3)=C(1)\REM       max. equiv. C                        (L-coupling)
50 GOSUB 1000\REM get next L value
55 GOSUB 1950\REM calculate frequencies
57 IF Z=0 AND F(3)<FMIN THEN 50
59 IF Z=0 THEN GOSUB 3000\GO TO 50
61 Z=0\A1=F(2)\A3=F(4)\GOSUB 2000
62 IF F(2)>FMAX THEN 500
```

T65-4-3

```
63 GOSUB 1000
65 GOSUB 1950
66 IF Z<>0 THEN 61
67 IF F(1)<A1 AND F(3)<A3 THEN Z=0\GO TO 63
69 GOSUB 3000\GO TO 63
180 GO TO 50
500 IF D$="A" THEN D$="P"\GO TO 29
510 PRINT #1\PRINT #1,U2$\PRINT #1,U2$\CLOSE #1%
520 PRINT\Q=0\INPUT"(1 = YES) Generate another file";Q
540 IF Q=1 THEN 26
560 GO TO 10000
1000 REM  ############# Inductor Table Subroutine ###############
1020 J=J-1
1030 IF J=0 AND M=-2 THEN 500
1040 IF J<1 THEN J=J+12\M=M-1
1060 L=V(J)*(10^M)
1100 RETURN
1500 REM  ######## Tap & Coupling Inductors Subroutine #########
1520 G=J\MM=M
1540 IF G<10 THEN G=J+3\MM=M-1\GO TO 1580
1560 G=J-9
1580 L2=V(G)*(10^MM)\REM Tap L
1600 G=J\MM=M
1620 IF G<6 THEN G=J+7\GO TO 1660
1640 G=J-5\MM=M+1
1660 L3=V(G)*(10^MM)\REM Coupling L
1680 RETURN
1800 REM  ####### Design Parameter Print-out Subroutine ########
1805 PRINT #1
1807 IF D$="P" THEN 1880
1810 PRINT #1,"  This is data file: ";S$
1815 PRINT #1,"   ( ";FMIN;" to ";FMAX;" kHz)"\PRINT #1
1820 PRINT #1,"  The following calculations are based upon:"
1840 PRINT #1,"     Tuning C = ";CMIN;" to ";CMAX;" pF"
1860 PRINT #1,"  Coupling C = ";CC;" pF  *  Stray C = ";CS;" pF"
1870 PRINT #1\PRINT #1, "    Active Configuration"\GO TO 1900
1880 PRINT #1, "    Passive Configuration"
1900 PRINT #1
1912 U$="  Main L       Freq.Range, kHz       Freq.Range, kHz     Tap L"
1913 IF D$="P" THEN U$=U$+"      Coupling L"
1914 PRINT #1, U$
1915 U$="     uH          (C coupling)         (L coupling)          uH"
1916 IF D$="P" THEN U$=U$+"           uH"
1917 PRINT #1, U$
1920 UU=3+LEN(U$)
1921 U$=""
1922 FOR I=1 TO UU
1923 U$=U$+"-"
1924 NEXT I
1930 PRINT #1, U$
1940 RETURN
1950 REM  ################# Math Subroutine ###################
1952 IF D$="A" THEN 1970
1955 REM                                PASSIVE
1957 F(1)=K/((L*C(1))^.5)\REM           Shorted Output (C-coupled)
1959 F(2)=K/((L*1.2*C(2))^.5)\REM       Open Output    (C-coupled)
1961 F(3)=K/((L*0.8*C(3))^.5)\REM       Shorted Input & Output
1963 F(4)=K/((L*1.2*C(4))^.5)\REM       Open Input & Output
1965 GO TO 1980
1970 REM                                ACTIVE
1972 F(1)=K/((L*1.2*C(1))^.5)\REM       (C-coupled)
1974 F(2)=K/((L*1.2*C(2))^.5)\REM       (C-coupled)
1976 F(3)=K/((L*C(3))^.5)\REM           Shorted Input
1978 F(4)=K/((L*1.2*C(4))^.5)\REM       Open Input
1980 RETURN
2000 REM  ############## Main Print-Out Subroutine ##############
2010 IF T=0 THEN GOSUB 1800
2020 T=1
2030 GOSUB 1500
2040 X=L\GOSUB 2300
2050 U$=U1$
2060 U1$=" ######.# 'L######.#"
2070 U$=U$+U1$+U1$+" "
2080 X=L2\GOSUB 2300
2090 U$=U$+U1$+" "
2095 IF D$="A" THEN 2130
2100 X=L3\GOSUB 2300
2110 U$=U$+U1$
2120 PRINT #1% USING U$, L;F(1);"-";F(2);F(3);"-";F(4);L2;L3\GO TO 2200
2130 PRINT #1% USING U$, L;F(1);"-";F(2);F(3);"-";F(4);L2
2200 RETURN
2300 REM  ############# Number-Formatting Subroutine ###############
2310 U1$="#######    "
2320 IF X<10 AND X>=1 THEN U1$="#######.# "
2330 IF X<1 AND X>=.1 THEN U1$="#######.##"
2340 IF X<.1 THEN U1$="####.#####"
2370 RETURN
3000 REM  ################## Drop-Back Subroutine ###################
3020 Z=Z+1
3040 IF J<11 THEN J=J+2\GO TO 3100
3060 J=J-10\M=M+1
3100 RETURN
10000 END
```

(end of program listing)

Sample Output File Print-Outs

This is data file: LWCOILS.TXT
(140 to 560 kHz)

The following calculations are based upon:
Tuning C = 10 to 365 pF
Coupling C = 62 pF * Stray C = 40 pF

Active Configuration

Main L uH	Freq.Range, kHz (C coupling)		Freq.Range, kHz (L coupling)		Tap L uH
3900	127.7 -	191.6	139.9 -	251.6	680
1800	187.9 -	282.0	205.9 -	370.3	330
820	278.5 -	417.8	305.0 -	548.7	150
390	403.8 -	605.8	442.3 -	795.6	68

****

Passive Configuration

Main L uH	Freq.Range, kHz (C coupling)		Freq.Range, kHz (L coupling)		Tap L uH	Coupling L uH
5600	116.7 -	159.9	130.5 -	210.0	1000	22000
3300	152.1 -	208.2	170.0 -	273.5	560	12000
1800	205.9 -	282.0	230.2 -	370.3	330	6800
1000	276.2 -	378.3	308.8 -	496.9	180	3900
560	369.1 -	505.5	412.7 -	664.0	100	2200
330	480.8 -	658.5	537.6 -	864.9	56	1200

****
****

TC5-4-4

This is data file: MWCOILS.TXT
(475 to 1900 kHz)

The following calculations are based upon:
Tuning C = 10 to 365 pF
Coupling C = 56 pF * Stray C = 40 pF

Active Configuration

Main L uH	Freq.Range, kHz (C coupling)	Freq.Range, kHz (L coupling)	Tap L uH
390	403.8 - 618.5	442.3 - 795.6	68
180	594.3 - 910.4	651.1 - 1171.1	33
82	880.5 - 1348.8	964.6 - 1735.2	15
39	1276.8 - 1955.8	1398.7 - 2516.0	6.8

****

Passive Configuration

Main L uH	Freq.Range, kHz (C coupling)	Freq.Range, kHz (L coupling)	Tap L uH	Coupling L uH
470	402.9 - 563.4	450.5 - 724.8	82	1800
270	531.6 - 743.3	594.3 - 956.2	47	1000
150	713.2 - 997.3	797.4 - 1282.9	27	560
82	964.6 - 1348.8	1078.4 - 1735.2	15	330
47	1274.1 - 1781.6	1424.5 - 2291.9	8.2	180
27	1681.0 - 2350.5	1879.4 - 3023.9	4.7	100

****
****

This is data file: TBCOILS.TXT
(1575 to 6300 kHz)

The following calculations are based upon:
Tuning C = 10 to 365 pF
Coupling C = 47 pF * Stray C = 40 pF

Active Configuration

Main L uH	Freq.Range, kHz (C coupling)	Freq.Range, kHz (L coupling)	Tap L uH
33	1388.0 - 2197.2	1520.5 - 2735.2	5.6
15	2058.8 - 3258.9	2255.3 - 4057.0	2.7
6.8	3057.8 - 4840.2	3349.6 - 6025.5	1.2
3.3	4389.4 - 6948.1	4808.3 - 8649.5	0.56

****

Passive Configuration

Main L uH	Freq.Range, kHz (C coupling)	Freq.Range, kHz (L coupling)	Tap L uH	Coupling L uH
39	1398.7 - 2021.1	1563.8 - 2516.0	6.8	150
22	1862.3 - 2691.0	2082.1 - 3349.9	3.9	82
12	2521.5 - 3643.6	2819.1 - 4535.8	2.2	47
6.8	3349.6 - 4840.2	3745.0 - 6025.5	1.2	27
3.3	4808.3 - 6948.1	5375.9 - 8649.5	0.56	12

****
****